Independent subsets of a finite set

Jochen Ziegenbalg
ziegenbalg.edu@gmail.com

In probability theory the *probabilistic* or *stochastic* independence of two subsets A and B of some set Ω frequently is defined by the equation

$$P(A \cap B) = P(A) \cdot P(B) \tag{1}$$

where P is the probability function (probability measure) of the probability space under consideration.

In the following we will consider subsets A and B of a finite n-element set Ω. We will use $P(X) := \frac{|X|}{|\Omega|} = \frac{|X|}{n}$ as a "probability" function and in analogy to (1) define the subsets A and B to be *independent* if the following equation holds:

$$\frac{|A \cap B|}{n} = \frac{|A|}{n} \cdot \frac{|B|}{n} \tag{2}$$

Whenever any problem about integer sequences is concerned, nowadays, the *Online Encyclopedia of Integer Sequences* (OEIS) is the place to look for information. The sequence of the numbers satisfying criterion (2) is discussed in the sequence OEIS A121312.

In OEIS A158345, the discussion is focussed on the case where A and B are *proper* subsets of Ω; i.e. $A \subsetneqq \Omega$ and $B \subsetneqq \Omega$. In this text, we will refer to this case as the "proper subsets case".

Pairs of independent nontrivial subsets

A further interesting specification might be the case where the subsets A and B are required to be *nontrivial*, i.e. $\emptyset \subsetneqq A \subsetneqq \Omega$ and $\emptyset \subsetneqq B \subsetneqq \Omega$ or, in other words, $1 \leq |A| \leq n-1$ and $1 \leq |B| \leq n-1$. This case will be referred to as the "nontrivial subsets case".

According to this terminology, the empty set $\emptyset$ is a trivial and proper subset and the full set Ω is a trivial but not a proper subset of Ω

In the following text let $a := |A|$, $b := |B|$ and $d := |A \cap B|$. Then the crucial condition for the independence of the sets A and B is

$$\frac{a}{n} \cdot \frac{b}{n} = \frac{d}{n} \tag{3}$$

or

$$a \cdot b = d \cdot n \tag{4}$$

Some consequences of equation (4):
* For $n = 0$ (i.e. $\Omega = \emptyset$) equation (4) is true. The only subset of Ω is Ω itself.
* If n is prime, then in the "nontrivial subsets case" there are no independent subsets A and B.
* For a better chance of the subsets being independent, it is favorable if n has many

divisors. If furthermore n is relatively small then the chance of the subsets being independent is even better (see Figure 1).

Remark: In the following we consider the "nontrivial subsets case" (i.e. $1 \leq |A| \leq n-1$ and $1 \leq |B| \leq n-1$). As is often the case, the problem can be tackled by a stochastic simulation for heuristic reasons. In the following figures, resulting from a simulation, the horizontal axis represents the cardinality n of Ω and the vertical axis represents the probability for randomly selected pairs of *nontrivial* subsets of Ω being independent. For more details of the simulation see [ZJ - Maxima Program].



Figure 1: Siimulation with random subsets



Figure 2: Simulation enhanced by the graph of $f(x) = \frac{2}{x}$

* Of course, the figures only provide an approximative empirical picture. But certain observations are quite suggestive: The larger the cardinality of Ω is, the smaller, by trend, is the chance for the independence of two subsets of Ω.
* If n is prime then the simulation dots seem to lie on the x-axis (see Figure 1).
* The simulation dots all seem to be below the graph of the function $f(x) = \frac{2}{x}$ (see Figure 2).

An *Example*: $n = 4$, $\Omega = \{1, 2, 3, 4\}$. Without the trivial subsets $\emptyset$ und Ω, there are the following 24 pairs of independent subsets:

$$\begin{array}{l}
\{\{1,2\},\{1,3\}\},\{\{1,2\},\{1,4\}\},\{\{1,2\},\{2,3\}\},\{\{1,2\},\{2,4\}\},\\
\{\{1,3\},\{1,2\}\},\{\{1,3\},\{1,4\}\},\{\{1,3\},\{2,3\}\},\{\{1,3\},\{3,4\}\},\\
\{\{1,4\},\{1,2\}\},\{\{1,4\},\{1,3\}\},\{\{1,4\},\{2,4\}\},\{\{1,4\},\{3,4\}\},\\
\{\{2,3\},\{1,2\}\},\{\{2,3\},\{1,3\}\},\{\{2,3\},\{2,4\}\},\{\{2,3\},\{3,4\}\},\\
\{\{2,4\},\{1,2\}\},\{\{2,4\},\{1,4\}\},\{\{2,4\},\{2,3\}\},\{\{2,4\},\{3,4\}\},\\
\{\{3,4\},\{1,3\}\},\{\{3,4\},\{1,4\}\},\{\{3,4\},\{2,3\}\},\{\{3,4\},\{2,4\}\}
\end{array}$$

The following Maxima program gives the number of pairs of independent nontrivial subsets of an n-element set.

Maxima program: pairs of all independent nontrivial subsets

```
pairs_independent_nontrivial_subsets(n) :=
  block([a, b, d, s : 0],
    for a:1 thru n-1 do
      for d:1 thru a do
        (b : n*d / a,
         if integerp(b) and b<n
           then
             s : s + binomial(n,a) *
                       binomial(a,d) *
                         binomial(n-a,b-d) ),
    s) ;
```

Example:

```
pairs_independent_nontrivial_subsets(4);
24
```

The following list `L30` is an initial interval (from 0 to 30) of the integers. By use of the `map`-command of Maxima the corresponding numbers of independent sets are computed.

```
L30 : makelist(i, i, 0, 30);
L_nontrivial_30 :
    map(pairs_independent_nontrivial_subsets, L30);

[0, 0, 0, 0, 24, 0, 720, 0, 7000, 15120, 126000, 0, 1777776, 0,
23543520, 55855800, 274565720, 0, 5337775872, 0, 63026049424,
117920013120, 995265791520, 0, 15265486117744, 14283091977000,
216344919117600, 240142901941800, 2854493961432480, 0,
55689696384165720]
```

A syntactically slightly different version of program is given here:

Maxima program: pairs of all independent nontrivial subsets (version 2)

```
a(n) :=
  sum(
    sum(
        (b : n*d / a,
         if integerp(b) and b<n
           then binomial(n,a) *
                  binomial(a,d) *
                    binomial(n-a,b-d)
           else 0),
      d,1,a),
   a,1,n-1) ;
```

Example:

```
a(6);
720
```

As is to be expected, the list returned by the call `map(a, L30)` is the same list as `L_nontrivial_30`. This list can be found in OEIS A340135.

Pairs of independent proper subsets

In OEIS A158345 the case of the *proper* subsets is discussed. Proper subsets of a set S, in this sense, are all subsets except for S itself. The first elements of this sequence are:

```
[1, 5, 13, 53, 61, 845, 253, 7509, 16141, 128045, 4093,
 1785965, 16381, 23576285, 55921333, 274696789, 262141,
 5338300157, 1048573, 63028146573, 117924207421, 995274180125,
 16777213, 15265519672173, 14283159085861 ];
```

If the *proper* subsets of a set Ω are to be considered, then the following pairs have to be added to the nontrivial subsets case:

$(\emptyset, U)$ with $U \subsetneqq \Omega$; these are $2^n - 1$ cases;

$(V, \emptyset)$ with $V \subsetneqq \Omega$; these are $2^n - 1$ cases.

With this way of counting, however, the pair $(\emptyset, \emptyset)$ is counted twice. Correcting this, the number of proper independent subsets of an n-element set ist given by the following program (using `a(n)` from above).

Maxima program: pairs of all independent proper subsets

```
pairs_independent_proper_subsets(n) :=
   if is(n=0) then 0 else a(n) + 2*(2^n -1) - 1
```

Applying this program to the list `L30` gives:

```
L_proper_30  :  map(pairs_independent_proper_subsets, L30);

[0, 1, 5, 13, 53, 61, 845, 253, 7509, 16141, 128045, 4093,
1785965, 16381, 23576285, 55921333, 274696789, 262141,
```

```
5338300157, 1048573, 63028146573, 117924207421, 995274180125,
16777213, 15265519672173, 14283159085861, 216345053335325,
240143170377253, 2854494498303389, 1073741821,
5568969853164936 5]
```

which contains the terms in OEIS A158345.

Pairs of all independent subsets

In OEIS A121312 the case of *all* independent subsets of a finite set is discussed. The first elements of this sequence are:

1, 4, 12, 28, 84, 124, 972, 508, 8020, 17164, 130092, 8188, 1794156, 32764, 23609052, 55986868, 274827860, 524284, 5338824444, 2097148, 63030243724, 117928401724, 995282568732, 33554428, 15265553226604, 14283226194724

If *all* subsets are to be considered, then the following pairs have to be added to the nontrivial subsets case:

$(\emptyset, X)$ with $X \subseteq \Omega$; these are 2^n cases
$(Y, \emptyset)$ with $Y \subseteq \Omega$; these are 2^n cases
(Ω, U) with $U \subseteq \Omega$; these are 2^n cases
(V, Ω) with $V \subseteq \Omega$; these are 2^n cases

With this way of counting, however, the four pairs $(\emptyset,\emptyset)$, $(\emptyset,\Omega)$, $(\Omega,\emptyset)$, (Ω,Ω) are counted twice. Correcting this, the number of nontrivial independent subsets of an n-element set ist given by the following program:

Maxima program: pairs of all independent subsets

```
pairs_independent_subsets(n) :=
  if is(n=0) then 1 else a(n) + 4*(2^n -1)
```

Example:

```
pairs_independent_subsets(6);
972
```

Application of this program to the list `L30` results in the list:

```
[1, 4, 12, 28, 84, 124, 972, 508, 8020, 17164, 130092, 8188,
1794156, 32764, 23609052, 55986868, 274827860, 524284,
5338824444, 2097148, 63030243724, 117928401724, 995282568732,
33554428, 15265553226604, 14283226194724, 216345187553052,
240143438812708, 2854495035174300, 2147483644,
55689700679133012]
```

and this is compatible with OEIS A121312.

Finally, in Table 1 some initial values (from 1 to 30) of the discussed functions are shown.

References

OEIS A121312: Online Encyclopedia of Integer Sequences;
https://oeis.org/A121312
Benoit Rittaud: Number of pairs of probabilistically independent subsets in a set composed of n elements.

OEIS A158345: Online Encyclopedia of Integer Sequences;
https://oeis.org/A158345
Harmen Wassenaar: The number of pairs of independent outcomes when rolling an n-sided die. Or in other terms, the number of pairs of proper subsets A, B of a set S, such that `#A/#S * #B/#S = #(A \intersect B)/#S`.

OEIS A340135: Online Encyclopedia of Integer Sequences;
https://oeis.org/A340135
Jochen Ziegenbalg: Number of pairs of independent nontrivial subsets of a finite set composed of n elements.

Ziegenbalg Jochen (Maxima worksheet saved as pdf): Independent subsets of a finite set
https://jochen-ziegenbalg.github.io/materialien/Maxima/independent-subsets-Maxima-prog.pdf

Ziegenbalg Jochen: Bedingte Wahrscheinlichkeiten, stochastische Unabhängigkeit, Mengendiagramme, Baumdiagramme und Vierfeldertafeln
(Some experimental and teaching materials - in German)
https://jochen-ziegenbalg.github.io/materialien/Manuskripte/Disko.pdf

Table of various independent subset types

n	independent pairs of all subsets	independent pairs of proper subsets	independent pairs of nontrivial subsets
0	1	0	0
1	4	1	0
2	12	5	0
3	28	13	0
4	84	53	24
5	124	61	0
6	972	845	720
7	508	253	0
8	8020	7509	7000
9	17164	16141	15120
10	130092	128045	126000
11	8188	4093	0
12	1794156	1785965	1777776
13	32764	16381	0
14	23609052	23576285	23543520
15	55986868	55921333	55855800
16	274827860	274696789	274565720
17	524284	262141	0
18	5338824444	5338300157	5337775872
19	2097148	1048573	0
20	63030243724	63028146573	63026049424
21	117928401724	117924207421	117920013120
22	995282568732	995274180125	995265791520
23	33554428	16777213	0
24	15265553226604	15265519672173	15265486117744
25	14283226194724	14283159085861	14283091977000
26	216345187553052	216345053335325	216344919117600
27	240143438812708	240143170377253	240142901941800
28	2854495035174300	2854494498303389	2854493961432480
29	2147483644	1073741821	0
30	55689700679133012	55689698531649365	55689696384165720]

Table 1: independent subsets of a finite n-element set